program ::= **class** **id** **{** *variable_declarations* *method_declarations* **}**

variable_declarations ::= *type* *variable_list* **;** *variable_declarations* | **ε**

type ::= **int** | **real**

variable_list ::= *variable* *more_variables*

more_variables ::= **,** *variable_list* | **ε**

variable ::= **id** *opt_array*

opt_array ::= **[** **num** **]** | **ε**

method_declarations ::= *method_declaration* *more_method_declarations*

more_method_declarations ::= *method_declaration* *more_method_declarations* | **ε**

method_declaration ::= **static** *method_return_type* **id** **(** *parameters* **)**
{ *variable_declarations* *statement_list* **}**

method_return_type ::= *type* | **void**

parameters ::= *parameter_list* | **ε**

parameter_list ::= *type* **id** *more_parameters*

more_parameters ::= **,** *parameter_list* | **ε**

statement_list ::= *statement* *statement_list* | **ε**

statement ::= *variable_loc* **=** *expression* **;**
| **id** **(** *expression_list* **)** **;**
| **if** **(** *expression* **)** *statement_block* *optional_else*
| **for** **(** *variable_loc* = *expression* **;** *expression* **;** *incr_decr_var* **)** *statement_block*
| **return** *optional_expression* **;**
| **break** **;**
| **continue** **;**
| *incr_decr_var* **;**
| *statement_block*

optional_expression ::= *expression* | **ε**

statement_block ::= **{** *statement_list* **}**

incr_decr_var ::= *variable_loc* **incdecop**

optional_else ::= **else** *statement_block* | ε

expression_list ::= *expression* *more_expressions* | **ε**

more_expressions ::= **,** *expression* *more_expressions* | **ε**

expression ::= *simple_expression* *optional_relop*

optional_relop ::= **relop** *simple_expression* | **ε**

simple_expression ::= *sign* *term* *optional_addops* | *term* *optional_addops*

optional_addops ::= **addop** *term* *optional_addops* | **ε**

term ::= *factor* *optional_mulop*

optional_mulop ::= **mulop** *term* | **ε**

factor ::= variable_loc | **id** **(** *expression_list* **)** | **num** | **(** *expression* **)** | **!** *factor*

variable_loc ::= **id** *opt_index*

opt_index ::= **[** *expression* **]** | **ε**

sign ::= + | -